// File: contracts/DeedHashedCloneFactory.sol



//SPDX-License-Identifier: Unlicense
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/proxy/Clones.sol";

interface IClonableDeedHashed {
  function initialize(address _admin) external;
  function update(bytes32 _hash, bytes32 _metahash) external;
  function transferOwnership(address newOwner) external;
  function getType() external pure returns(bytes32);
}

contract DeedHashedCloneFactory is Ownable {

    event NewDeedHashedClone(
      address indexed referenceContract,
      address indexed cloneContract,
      address indexed admin
    );

    event UpdatedDeedHashedClone(
      address indexed cloneContract,
      bytes32 indexed _hash,
      bytes32 indexed _metahash
    );

    event TransferOwnershipDeedHashedClone(
      address indexed cloneContract,
      address _newOwner
    );

    event SetClonableDeedHashedReferenceContractValidity(
      address indexed referenceContract,
      bool validity
    );

    mapping(address => bool) public validClonableDeedHashed;
    mapping(address => bool) public isClone;

    constructor(
      address _clonableDeedHashedReference
    ) {
      validClonableDeedHashed[_clonableDeedHashedReference] = true;
      emit SetClonableDeedHashedReferenceContractValidity(_clonableDeedHashedReference, true);
    }

    function setClonableDeedHashedReferenceValidity(
      address _clonableDeedHashedReference,
      bool _validity
    ) external onlyOwner {
      validClonableDeedHashed[_clonableDeedHashedReference] = _validity;
      emit SetClonableDeedHashedReferenceContractValidity(_clonableDeedHashedReference, _validity);
    }

    function newDeedHashedClone(
      address _clonableDeedHashedReference
    ) external onlyOwner {
      require(validClonableDeedHashed[_clonableDeedHashedReference], "INVALID_WHITELIST_REFERENCE_CONTRACT");
      // Deploy new DeedHashed contract
      address newDeedHashedCloneAddress = Clones.clone(_clonableDeedHashedReference);
      isClone[newDeedHashedCloneAddress] = true;
      IClonableDeedHashed deedHashedClone = IClonableDeedHashed(newDeedHashedCloneAddress);
      deedHashedClone.initialize(address(this));
      emit NewDeedHashedClone(_clonableDeedHashedReference, newDeedHashedCloneAddress, address(this));
    }

    function updateDeedHashedClone(
      address _deedHashedClone,
      bytes32 _hash,
      bytes32 _metahash
    ) external onlyOwner {
      require(isClone[_deedHashedClone] == true, "INVALID_CLONE_ADDRESS");
      IClonableDeedHashed deedHashedClone = IClonableDeedHashed(_deedHashedClone);
      deedHashedClone.update(_hash, _metahash);
      emit UpdatedDeedHashedClone(_deedHashedClone, _hash, _metahash);
    }

    function transferDeedHashedCloneOwnership(
      address _deedHashedClone,
      address _newOwner
    ) external onlyOwner {
      require(isClone[_deedHashedClone] == true, "INVALID_CLONE_ADDRESS");
      IClonableDeedHashed deedHashedClone = IClonableDeedHashed(_deedHashedClone);
      deedHashedClone.transferOwnership(_newOwner);
      emit TransferOwnershipDeedHashedClone(_deedHashedClone, _newOwner);
    }

    function getTypeDeedHashedClone(
      address _deedHashedClone
    ) external view returns (bytes32) {
      require(isClone[_deedHashedClone] == true, "INVALID_CLONE_ADDRESS");
      IClonableDeedHashed deedHashedClone = IClonableDeedHashed(_deedHashedClone);
      return deedHashedClone.getType();
    }

}
// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/proxy/Clones.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (proxy/Clones.sol)

pragma solidity ^0.8.0;

/**
 * @dev https://eips.ethereum.org/EIPS/eip-1167[EIP 1167] is a standard for
 * deploying minimal proxy contracts, also known as "clones".
 *
 * > To simply and cheaply clone contract functionality in an immutable way, this standard specifies
 * > a minimal bytecode implementation that delegates all calls to a known, fixed address.
 *
 * The library includes functions to deploy a proxy using either `create` (traditional deployment) or `create2`
 * (salted deterministic deployment). It also includes functions to predict the addresses of clones deployed using the
 * deterministic method.
 *
 * _Available since v3.4._
 */
library Clones {
    /**
     * @dev Deploys and returns the address of a clone that mimics the behaviour of `implementation`.
     *
     * This function uses the create opcode, which should never revert.
     */
    function clone(address implementation) internal returns (address instance) {
        /// @solidity memory-safe-assembly
        assembly {
            let ptr := mload(0x40)
            mstore(ptr, 0x3d602d80600a3d3981f3363d3d373d3d3d363d73000000000000000000000000)
            mstore(add(ptr, 0x14), shl(0x60, implementation))
            mstore(add(ptr, 0x28), 0x5af43d82803e903d91602b57fd5bf30000000000000000000000000000000000)
            instance := create(0, ptr, 0x37)
        }
        require(instance != address(0), "ERC1167: create failed");
    }

    /**
     * @dev Deploys and returns the address of a clone that mimics the behaviour of `implementation`.
     *
     * This function uses the create2 opcode and a `salt` to deterministically deploy
     * the clone. Using the same `implementation` and `salt` multiple time will revert, since
     * the clones cannot be deployed twice at the same address.
     */
    function cloneDeterministic(address implementation, bytes32 salt) internal returns (address instance) {
        /// @solidity memory-safe-assembly
        assembly {
            let ptr := mload(0x40)
            mstore(ptr, 0x3d602d80600a3d3981f3363d3d373d3d3d363d73000000000000000000000000)
            mstore(add(ptr, 0x14), shl(0x60, implementation))
            mstore(add(ptr, 0x28), 0x5af43d82803e903d91602b57fd5bf30000000000000000000000000000000000)
            instance := create2(0, ptr, 0x37, salt)
        }
        require(instance != address(0), "ERC1167: create2 failed");
    }

    /**
     * @dev Computes the address of a clone deployed using {Clones-cloneDeterministic}.
     */
    function predictDeterministicAddress(
        address implementation,
        bytes32 salt,
        address deployer
    ) internal pure returns (address predicted) {
        /// @solidity memory-safe-assembly
        assembly {
            let ptr := mload(0x40)
            mstore(ptr, 0x3d602d80600a3d3981f3363d3d373d3d3d363d73000000000000000000000000)
            mstore(add(ptr, 0x14), shl(0x60, implementation))
            mstore(add(ptr, 0x28), 0x5af43d82803e903d91602b57fd5bf3ff00000000000000000000000000000000)
            mstore(add(ptr, 0x38), shl(0x60, deployer))
            mstore(add(ptr, 0x4c), salt)
            mstore(add(ptr, 0x6c), keccak256(ptr, 0x37))
            predicted := keccak256(add(ptr, 0x37), 0x55)
        }
    }

    /**
     * @dev Computes the address of a clone deployed using {Clones-cloneDeterministic}.
     */
    function predictDeterministicAddress(address implementation, bytes32 salt)
        internal
        view
        returns (address predicted)
    {
        return predictDeterministicAddress(implementation, salt, address(this));
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}